
Before Getting Started

It may be difficult to identify the important points in some illustrations without a
color copy of this booklet.  You can print or read this booklet in color from
MicroImages’ web site.  The web site is also your source for the newest Getting
Started booklets on other topics.  You can download an installation guide, sample
data, and the latest version of TNTlite.

http://www.microimages.com

Some mapping projects may require the use of specialized symbols for lines and
points in vector and CAD objects.  The cartographic scripting language in TNT-
mips®, TNTview®, and TNTedit™ provides a complete and flexible set of drawing
functions that allow you to design custom map symbols for many applications.
CartoScripts extend the symbol-creation capabilities found in the standard point
and line style editors in the TNT products.  The exercises in this booklet intro-
duce the most commonly-used CartoScript functions, and provide many sample
CartoScripts for point and line symbols.

Prerequisite Skills This booklet assumes that you have completed the exercises
in the following Getting Started booklets: Displaying Geospatial Data, Navigat-
ing, Creating and Using Styles, and Building and Using Queries.  Those exercises
introduce essential skills and basic techniques that are not covered again here.
Please consult those booklets and the TNTmips reference manual for any review
you need.

Sample Data  The exercises presented in this booklet use sample data that is
distributed with the TNT products.  If you do not have access to a TNT products
CD, you can download the data from MicroImages’ web site.  In particular, this
booklet uses the CARTOSMP, TOWNS and GGMAP Project Files in the CARTOSCR data
collection.  Install the sample files on your hard drive so changes can be saved as
you work with them.

More Documentation  This booklet is intended only as an introduction to using
CartoScripts to style vector or CAD elements.  For more information consult the
Display Volume of the TNTmips reference manual.

TNTmips and TNTlite®  TNTmips comes in two versions: the professional ver-
sion and the free TNTlite version. This booklet refers to both versions as
“TNTmips.”  If you did not purchase the professional version (which requires a
hardware key), TNTmips operates in TNTlite mode, which limits object size and
does not allow export.  All of the exercises can be completed in TNTlite using the
sample geodata provided.

Randall B. Smith, Ph.D., 28 August 2002
©MicroImages, Inc., 1999

Welcome to Using CartoScripts

STEPS
� make sure the sample

files mentioned on page
2 have been copied to
your hard drive

� start TNTmips
� choose Display / Spatial

Data from the main
menu

The style editors in TNTmips and TNTview allow
you to select, modify, combine, or create a wide va-
riety of standard point and line symbols, as described
in Getting Started: Creating and Using Styles.  For
those instances in which the standard style editors
cannot provide the appropriate symbol, you can use
CartoScripts™ to design custom map symbols for
points and lines.

CartoScripts are style scripts that utilize special func-
tions in the TNTmips Query language (in the
Cartoscripts function group). The CartoScript func-
tions allow you to draw and navigate along line
elements and to draw new lines and shapes to form
symbols.  Symbols can be repeated along line ele-
ments or drawn singly for point elements. You can
add labels to the symbols using text from associated
database tables and optimize label placement within
a single drawing layer.  You can also structure a script
to use element attributes to vary the symbol styling.

The exercises in this booklet introduce and explain
the most commonly-used CartoScript functions and
show how you can structure scripts to produce vari-
ous effects for point and line symbols.  CartoScripts
are subject to the same syntax rules as standard
database queries and SML scripts.  For a review of
basic query syntax, consult Getting Started: Building
and Using Queries.

Geologic maps are one example of maps requiring
specialized point and line symbols that can be drawn
using CartoScripts.  Point symbols are used to indi-
cate the orientation of outcrop-scale structures, while
special line symbols are used to represent map-scale
features.  Several of the exercises in this booklet use
geological examples to illustrate elements of
CartoScript structure.

You can download additional sample CartoScripts for geological point and line
symbols from the MicroImages web site:

www.microimages.com / freestuf / cartoscripts

The exercises on pages 4-
19 illustrate the use of
CartoScript functions to
create point symbols.  Pages
4-8 introduce the basic
functions used to draw lines
and simple geometric
shapes.  Pages 9-19 lead
you through adding text
labels from database fields,
drawing more complex
symbols, varying symbol
orientation by attribute, and
optimizing label placement.

The exercises on pages 20-
33 show you how to use
CartoScripts to create line
symbols.  Basic line
navigation and drawing
functions are introduced on
pages 20-22.  Script
structures to create repeated
symbols and placement of
text labels for lines are
explained on pages 23-33.

Use of CartoScripts with
legends is discussed on
pages 34-37, and pages 38-
39 provide a complete list of
available CartoScript
functions.

Draw Simple Point Symbols
STEPS
� click the New 2D

Group icon
button on the Display
toolbar

� click on the Add
Vector icon button
in the Group Controls
window and choose Add
Vector Layer

� navigate to the CARTOSMP

Project File in the
CARTOSCR data collection
and select the SAMPLES

vector object
� click on the Points tab on

the Vector Layer
Controls window

� select By Script from the
Style option menu and
click [Specify...]

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object FLAGQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

## FlagQry
LineStyleSetColor(225,0,0);
LineStyleSetLineWidth(1);
LineStyleLineTo(90,20);
LineStyleLineTo(-30,8);
LineStyleLineTo(-150,8);
LineStyleMoveTo(-90,12);
LineStyleLineTo(0,8);

The CartoScript you open in this exercise
(shown in the box below) draws the flag
symbol shown to the right.  The first line
in the script is a comment line with the
script name (remember that comments are
preceded by the “#” character). The LineStyleSet-
Color( ) function in Line 2 sets Red, Green, and Blue
values (between 0 and 255) that determine the color
of lines drawn by the drawing functions that follow
(in this case, red lines).  The function in Line 3 sets
the line width for the drawing functions (more about
this below).

The remaining script lines actually draw the sym-
bol.  The LineStyleLineTo( ) function draws a line
to the point specified by a direction (first numeric
parameter) and distance (second parameter).  The
LineStyleMoveTo( ) function uses the same sequence
of parameters to move the “pen” location without
drawing.  Both functions reference a local coordi-
nate system centered on the current point element
(the script is read and evaluated once for each point
element in the object).  Directions for these func-
tions are specified by angles (0 to 180 and 0 to -180)
relative to the positive x axis of the object coordi-
nate system.  Object coordinates also provide the

default units for the distance parameters in
these functions, as well as for the width pa-
rameter in the LineStyleSetLineWidth( )
function.  The sequence of drawing actions
is illustrated below, with the dot indicating
the pen position at the end of each action.

1
2
3
4
5
6
7
8

LineTo(90,20)

LineTo(-30,8) LineTo(-150,8) MoveTo(-90,12) LineTo(0,8)

-90

STEPS
� click the Vector

icon button in the
Layer icon row to open
the Vector Layer
Controls window

� reopen the Query Editor
window for point styles

� delete the statement with
the LineStyleMoveTo( )
function and insert the
statements shown in
bold text below

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Using Anchors

LineStyleSetColor(225,0,0);
LineStyleSetLineWidth(1);
LineStyleDropAnchor(1);
LineStyleLineTo(90,20);
LineStyleLineTo(-30,8);
LineStyleLineTo(-150,8);
LineStyleMoveToAnchor(1);
LineStyleLineTo(0,8);
LineStyleMoveToAnchor(1);
LineStyleLineTo(-135,8);

In the sequence of drawing movements in the flag
script in the previous exercise, the pen position re-
turns to the origin of the local coordinate system
before drawing the final line at the base of the flag.
The simple geometry of the flag symbol makes it
relatively easy to calculate the angle and distance for
the LineStyleMoveTo( ) function in Line 7 that
moves the pen to the origin.  But you can avoid the
need for such calculations by using anchors: posi-
tions you record for later use in a set of drawing
actions. The LineStyleDropAnchor( ) function sets
an anchor position and assigns it the number you
enter as the numeric parameter for the function.  The
LineStyleMoveToAnchor( ) function moves the pen
to the specified anchor position.  There is also
a LineStyleLineToAnchor( ) function that
draws a line from the current pen position to
the specified anchor position.  You can es-
tablish multiple anchor points to aid in
drawing complex symbols, and use them in
any order.  In the script on this page, we place

an anchor at the origin before
beginning to draw, and return to
the anchor position twice more
to draw lines from the base of the
flag.

You can manually
enter  function
names, database
field names, symbol
names, or operators,
or use the options on
the Insert menu to
insert the required
items into the script.

Choose Cartoscript from the Function
Group menu to see a function list with
only the CartoScript style functions.

STEPS
� reopen the Query Editor

window for point styles
� edit the script to

duplicate the text below;
the new statement is in
bold type

Using Built-in Geometric Shapes

LineStyleDrawEllipse(0,0,8,3,30,0,0);

� reopen the Query Editor
window

� replace the last line in
the script with the
statement shown below

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Several functions that draw simple geometric shapes
are included in the CartoScript function set.  The
LineStyleDrawRectangle( ) and LineStyleDrawCir-
cle( ) functions draw their respective shape centered
on the current pen position, which is left unchanged.

The first script for this exercise draws a filled
rectangle at the base of the flag symbol.  Four
numeric parameters are used in this instance
to control the rectangle function: width,
height, angle, and dofill.  The first two pa-
rameters are required, and specify the width

and height of the rectangle.  The third
(optional) parameter specifies a rota-

tion angle, which in this example is equal to zero.
The last parameter (also optional)  determines wheth-
er the shape is filled with the current line color (1),
or left unfilled (0).

The second example in this exercise draws an un-
filled ellipse at the base of the flag symbol.  The
LineStyleDrawEllipse( ) function has up to 7 param-
eters: angle, distance, radius_x, radius_y, rotangle,
isAngleAbs, and dofill; the first four are required.
The initial angle and distance parameters allow you
to automatically move the pen to a new position be-
fore drawing.  In this example, both are set to zero,
leaving the ellipse centered on the base of the flag.
The pen returns to the ellipse center after drawing.

The ellipse dimensions are ini-
tially set parallel to the x and y

coordinate axes by the two radius parameters (in this
example, 8 and 3 units, respectively).  The value of
30 for the angle parameter rotates the ellipse 30 de-
grees counterclockwise.  The isAngleAbs parameter
determines whether the ellipse is drawn and rotated
relative to the local coordinate system (0), or rela-
tive to global object coordinates (1).  This distinction
does not exist for point data, but becomes important
when styling lines, as we will see later.

LineStyleSetColor(225,0,0);
LineStyleSetLineWidth(1);
LineStyleDropAnchor(1);
LineStyleLineTo(90,20);
LineStyleLineTo(-30,8);
LineStyleLineTo(-150,8);
LineStyleMoveToAnchor(1);
LineStyleDrawRectangle(10,5,0,1);

STEPS
� reopen the Query Editor

window for point styles
� edit the script to

duplicate the text below,
adding the statement
shown in bold type

Recording and Drawing Polygons

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

� reopen the Query Editor
window

� change the values for
the cap and join
parameters as shown,
and add the statements
shown in bold

� click [OK] in the Query
Editor window and
again in the Vector
Layer Controls
window

LineStyleSetColor(225,0,0);
LineStyleSetLineWidth(1);
LineStyleSetCapJoinType(1,1);
LineStyleLineTo(90,20);
LineStyleLineTo(-30,8);
LineStyleLineTo(-150,8);

LineStyleSetColor(225,0,0);
LineStyleSetLineWidth(1);
LineStyleSetCapJoinType(0,0);
LineStyleLineTo(90,20);
LineStyleDropAnchor(2);
LineStyleRecordPolygon(1);
LineStyleLineTo(-30,8);
LineStyleLineTo(-150,8);
LineStyleMoveToAnchor(2);
LineStyleDrawPolygon(1);

Styling for the ends of lines is set using the Line-
StyleSetCapJoinType( ) function, which has capstype
and jointype parameters.  The capstype parameter
draws square line ends when set to 1, or rounded
ends when set to 0.  The jointype parameter uses the
same values to determine styling for the
ends of segments of a polygon outline or
polyline.  The default value for both pa-
rameters is 0, so rounded ends are drawn
if you do not include this statement in a
script.

The script in the second half of this exercise draws a
solid-color flag symbol by drawing the triangular flag
element as a filled polygon.  You can draw simple or
complex polygon shapes by using functions to exe-
cute the following steps: 1) initiate recording of
vertex locations; 2) move to or draw lines to each
vertex location in turn; 3) connect the vertices to draw
the polygon.  The LineStyleRecordPolygon( ) func-
tion has a single start_stop parameter; a value of 1
starts recording vertex locations specified by pen
movements in subsequent statements.  The Line-
StyleDrawPolygon( ) function forms a polygon using
the recorded vertices, and has a single dofill param-
eter (set to 1 in this example to fill the triangle).  The
LineStyleDrawPolygon( ) function also stops the re-
cording of vertex
locations, so there is
no need to explicitly
stop recording with a
LineStyleRecord-
Polygon(0) statment
following the vertex
movement list.  You
can also use the same
structure to  record vertex locations to connect as
segments of a single line with the LineStyleDraw-
Polyline( ) function.

� select Close from the
Group menu when you
have completed this
exercise

Using 3D Shapes
CartoScript functions are also available to draw per-
spective renderings of simple three-dimensional
shapes: a rectangular solid, vertical cylinder, and
vertical cone.  Edge lines in each symbol are drawn
using the color specified by the LineStyleSetColor( )
function.  Three color parameters for each function
set the red, green, and blue values for the fill color.
The cube symbol is drawn with the base of the front
face centered on the current pen position.  The cyl-
inder and cone symbols are drawn so that the center
of the basal ellipse coincides with the current pen
position.

The width, depth, and height parameters for the Line-
StyleDrawCube( ) function specify the lengths of the
corresponding edges of the rectangular solid.  If you
want the symbol to appear in perspective as a true
cube, as in this exercise, the width and height should
be equal and the depth value should be about half
the length of the other dimensions.

You can define numeric or string variables in a script
for later use as function parameters.  Defining vari-
ables (with comments) at the beginning of a script
makes it easier to find and edit necessary parameter
values when you are reusing and modifying a script.

LineStyleDrawCone( )

LineStyleDrawCylinder( )

# CubeQry1
# Set dimensions of cube symbol
width = 15;
depth = 0.5 * width;
height = width;

# Set color for cube faces
red = 255;  green= 0;  blue = 0;

# Draw cube symbol
LineStyleSetColor(0,0,0);    # line color for edges
LineStyleSetLineWidth(0);
LineStyleDrawCube(width,depth,height,red,green,blue);

LineStyleDrawCube( )

STEPS
� click the Open

icon button on
the Spatial Data Display
toolbar and choose
Open Group from the
menu

� select the CUBEGROUP

object from the CARTOSMP

Project File
� click the Vector

icon button in the
Layer icon row for the
SAMPLES layer to open the
Vector Layer Controls
window

� open the Query Editor
window for point styles
and examine the script
CUBEQRY1

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Text Labels from Database Fields
STEPS
� click the Vector

icon button for the
SAMPLES layer to reopen
the Vector Layer
Controls window

� open the Query Editor
window for point styles

� choose Open / RVC
Object from the File
menu

� select object CUBEQRY2
in the CARTOSMP Project
File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

The script in this exercise adds a boxed text label to
the cube symbol.  The additional statements needed
to format and draw the label are shown below.  The
label is a sample number stored as a numeric value
in a database field.  In order to use it as a label, the
number must first be converted to a text string and
assigned to a string variable using the sprintf( ) func-
tion.  The first parameter in this function is a string
(in quotes) with formatting information; “%d” indi-
cates an integer value.  The second parameter in this
case is the location of the numeric value, specified
in the form TableName.FieldName.

Text labels are drawn with the lower left corner cor-
responding to the pen position.  The orientation of a
label is set by the angle parameter.  The border pa-
rameter specifies the width of the border between
the text label and its surrounding box.  The last pa-
rameter shown here, isAbs(0), indicates the reference
frame of the angle parameter.

# Read sample number from database field and
# convert to text string for use as a label
label$ = sprintf(“%d”,Samples.Number);

# String variable for label text font
font$ = “ARIALBD.TTF”;

# Define color variables for text
tred = 0; tgreen= 0; tblue = 0;

# Define fill color variables for text box
fillred = 255; fillgreen = 255; fillblue = 170;

# Define height, angle, and border width of text box
t_height = 10; angle = 0; border = 2;

# Set color and font for text label
LineStyleSetTextColor(tred,tgreen,tblue,fillred,

fillgreen,fillblue);
LineStyleSetFont(font$);

# Move pen to right of symbol and draw label
LineStyleMoveTo(0, width * 1.2);
LineStyleDrawTextBox(label$,t_height,angle,border,0);

You should get in the
habit of terminating each
statement in a script with
a semicolon (;). This
helps the syntax checker
pinpoint the location of
syntax errors.

Setting Coordinate Type Options
STEPS
� zoom in and out and

note the effect on the
relative sizes of point
symbols and contours

� reopen the Vector Layer
Controls window for the
SAMPLES layer and Query
Editor window for point
styles

� choose Open / RVC
Object from the File
menu

� select object CUBEQRY3
from the CARTOSMP

Project File

# CubeQry3
# Set dimensions of cube symbol
LineStyleSetCoordType(1);
width = 4;
depth = 0.5 * width;
height = width;

Different zoom levels with default scaling to
object coordinates.  The point symbol has a
constant size in object coordinates.

Different
zoom levels
with scaling
to millimeters.  The point symbol main-
tains a constant size in screen (or
print) coordinates.

The default units for the size and distance values you
use in CartoScripts are in internal object coordinates
(meters for the objects you have used in these exer-
cises).  Thus as you change zoom levels, the size of
the symbols on the screen change as the display scale
changes, maintaining a constant size in object coor-
dinates.  The script in this exercise adds the statement
shown in bold type to the previous CubeQry2 (along
with changes in the symbol and label size values).  A
parameter value of 1 for the LineStyleSetCoord-
Type( ) function changes the size and distance units
to millimeters at the current display scale or print

scale.  When you change zoom levels or
print scales, symbols maintain a constant
display size in millimeters.  This setting
works best with layouts designed only
for display  (A value of 0 for this pa-
rameter is equivalent to the default
condition).

If you are using CartoScripts to draw map elements
designed for printing at a particular scale (such as
1:24,000), you can use the default coordinate setting
to maintain the relative sizes of different elements at
different zoom levels for screen display, but use the
map scale to compute the element sizes needed to
produce the desired sizes on the final printed map.
See the script on pages 14-15 for an example of this
scaling approach.

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

� zoom in and out and note
the effect on the relative
sizes of  the symbols and
contour lines

� select Close from the
Group menu when you
have completed this
exercise, and click [No]
on the dialog when you
are asked
whether to
save changes
to the group

# Read values for element abundances from
# database and assign to variables to use
# for cylinder heights
cuVal = Geochemistry.Cu;
pbVal = Geochemistry.Pb;
znVal = Geochemistry.Zn;
# Set edge line color and cylinder dimensions
LineStyleSetColor(0,0,0);
longAxis = 120; shortAxis = 50;
# Set text color and font
LineStyleSetTextColor(0,0,0);
LineStyleSetFont(“ARIALBD.TTF”);
# Draw three cylinders side by side
LineStyleDropAnchor(1);
LineStyleDrawCylinder(longAxis,shortAxis,cuVal,255,0,0);
LineStyleMoveTo(0,longAxis); # move right by width of cylinder
LineStyleDrawCylinder(longAxis,shortAxis,pbVal,0,255,0);
LineStyleMoveTo(0,longAxis);
LineStyleDrawCylinder(longAxis,shortAxis,znVal,0,0,255);
# Draw label centered below each cylinder
LineStyleMoveToAnchor(1);     # move to base of first cylinder
LineStyleMoveTo(-90,100);     # move down to make room for label
LineStyleTextNextPosition(“Cu”,70,0,0,next_x,next_y,length);
LineStyleMoveTo(180,length * 0.4); # move label point to left
LineStyleDrawText(“Cu”,70,0,0); # to center first label
LineStyleMoveTo(0,longAxis);    # move right by width of cylinder
LineStyleDrawText(“Pb”,70,0,0);
LineStyleMoveTo(0,longAxis);
LineStyleDrawText(“Zn”,70,0,0);

3D Cylinder Bar Graph
The 3D symbols can be combined together to form
3D bar graphs, with the height of each bar deter-
mined by the value in a database field. You can
change the perspective of the cylinder and cone sym-
bols by varying the relative lengths of the long and
short axes of the cylinder and cone base.

The LineStyleTextNextPosition( ) function is used
in this script to approximately center the “Cu” ele-
ment label below its cylinder.  The first four
parameters of this function specify the string, its
height, angle, and local or absolute coordinate refer-
ence.  The last three parameters, next_x, next_y, and
length, are variables which the function creates to
hold the x-coordinate, y-coordinate, and length of
the label string. These values can be used in subse-
quent statements to guide positioning.

� remove the GEOCHEMVEC layer when
you have completed this exercise

STEPS
� click the New 2D

Group icon
button on the Display
toolbar

� click on the Add
Vector icon button
and choose Add Vector
Layer

� select the GEOCHEMVEC

object from the CARTOSMP

Project File
� set point styling to By

Script and open the
Query Editor window

� choose Open / RVC
Object from the File
menu

� select object CYLINGRAPH

in the CARTOSMP Project
File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Orienting Symbols by Attribute
You can use a script to draw symbols that vary in
orientation depending on a direction value read from
a database field.  In this instance the arrow direc-
tions are in azimuth form (0 to 360° angle measured
clockwise from north), and must be converted to the
internal coordinate system used by the drawing func-
tions. This script also converts all negative direction
values to the corresponding positive values, but this
conversion is not required.

The LineStyleDrawArrow( ) function draws arrow-
heads bounded by straight lines whose length is
defined by the headSize variable in this script.  The
angle between these bounding lines is set by the
sweepAngle parameter.  If the dofill parameter is set
to 1, the head is filled to form a solid triangle.  The
function updates the pen position  to the tip of the
arrow head.  Arrow heads do not render well with a
line width larger than 0, so this script draws the ar-
row with 0 line width, then redraws the arrow stem
with a wider line.

STEPS
� click on the Add

Vector icon button and
choose Add Vector
Layer

� select the ARROWPTS

object from the CARTOSMP

Project File
� set point styling to By

Script and open the
Query Editor window

� choose Open / RVC
Object from the File
menu

� select object ARROWQRY1
in the CARTOSMP Project
File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

# Read azimuth from database table
azim = Direct.Azimuth;
# Convert azimuth to internal coordinate system
direction = -(azim - 90);

if (direction < 0) then  direction = direction + 360;
# Set color values for symbol
red = 0; green = 0; blue = 0;
LineStyleSetColor(red,green,blue);
# Set dimensions for arrow
arrowLength = 30;
headSize = 0.4 * arrowLength;
sweepAngle = 40; dofill = 1;
# Draw arrow with zero line width, tip of stem at point
LineStyleDropAnchor(0); # anchor at point
LineStyleSetLineWidth(0);
LineStyleDrawArrow(direction,arrowLength,headSize,sweepAngle,dofill);
LineStyleDropAnchor(1); # anchor at tip of arrow
# Redraw arrow stem with wider line
stemLength = arrowLength - headSize * cosd(sweepAngle);
LineStyleMoveToAnchor(0);
LineStyleSetLineWidth(1.5);
LineStyleLineTo(direction,stemLength);
LineStyleMoveToAnchor(1); # pen to arrow tip in prep for label

headSize

sweepAngle

STEPS
� reopen the Query Editor

window for point styles
� choose Open / RVC

Object from the File
menu

� select object ARROWQRY2
in the CARTOSMP Project
File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Calculating Label Positions

# Convert azimuth value to text string for use as a label
label$ = sprintf(“%d”,azim);
# Find length of label text for label positioning
height = 10;
LineStyleTextNextPosition(label$,height,0,0,next_x,next_y,length);
# Set font name and color
LineStyleSetFont(“ARIALBD.TTF”);
LineStyleSetTextColor(red,green,blue);
# Compute label shift perpendicular to arrow to center label
shift1 = 0.5 * (height*cosd(direction) - length*sind(direction));
# Compute label shift parallel to arrow to avoid overwriting arrow
offset = arrowLength * 0.1;
if (direction >= 0 and direction < 90) then

shift2 = offset;
else if ( direction >= 90 and direction < 180 ) then

shift2 = offset - length * cosd(direction);
else if ( direction >= 180 and direction < 270 ) then

shift2 = offset - length * cosd(direction)
- height * sind(direction);

else if ( direction >= 270 and direction <= 360 ) then
shift2 = offset - height * sind(direction);

# shift pen position and draw label
if (shift1 < 0 ) {   # MoveTo distances can’t be less than 0

shift1 = abs(shift1);  # absolute value
LineStyleMoveTo(direction + 90,shift1); }
else LineStyleMoveTo(direction - 90,shift1);

LineStyleMoveTo(direction,shift2)
LineStyleDrawText(label$,height,0,0)

The script in this exercise adds a label with the azi-
muth to each arrow symbol from the previous
exercise.  The additional statements needed to for-
mat and draw the labels are shown below.  The tricky
part is varying the label position based on the arrow
orientation to avoid overlap between the arrow and
its label.  The script computes two pen shifts which
are applied before the label is drawn, as explained in
the comments.  Each is computed as a function of
the direction angle and the height and length of the
label string.  The second shift must be computed sep-
arately for each quadrant, being careful that the
resulting distance is a positive value (values for dis-
tance parameters of drawing functions must be
positive; a negative value is interpreted as equal to
0).

� remove the
ARROWPTS  layer
when you have
completed this exercise

# BeddingQry
# Read strike azimuth and dip value from table.field
azStrike = Bedding.Strike; dip1 = Bedding.Dip;
# Check logical field for overturned bedding.  Variable is set
# to 1 if Yes, 0 if No
overturned = Bedding.Overturned;
# Variables define the color of the symbols and label
red = 0; green = 0; blue = 0;
# This variable defines the denominator of the intended map scale.
scale = 5000;
# These variables define the dimensions and line widths of the
# symbol.  strikeLengthMap is the desired length of the symbol
# strike line in mm, assuming vector coordinates are in meters.
# lineWidthMap is the desired line width in mm.
strikeLengthMap = 6; lineWidthMap = 0.3;
strikeLength = strikeLengthMap * scale / 1000;
halfLength = 0.5 * strikeLength;
tickLength = halfLength / 3;
doubTick = tickLength * 2;
lineWidth = lineWidthMap * scale / 1000;

Drawing Strike and Dip Symbols
STEPS
� click on the Add

Vector icon button and
choose Add Vector Layer

� select the BEDDING object
from the CARTOSMP Project
File

� set point styling to By
Script and open the
Query Editor window

� choose Open / RVC
Object from the File
menu

� select object BEDDINGQRY

in the CARTOSMP Project
File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

� select Close from the
Group menu when you
have completed this
exercise

Inclined Overturned Horizontal Vertical

Geologic maps use special point symbols to indicate
the orientation of planar or linear structures in rock
outcrops.  The script in this exercise draws standard
symbols showing the strike and dip of layering (bed-
ding) in sedimentary rocks, as shown below.  It uses
many of the functions and concepts introduced pre-
viously.  The dip and strike are read from an
associated database table.  (The strike angle must be
specified as an azimuth using the so-called right-hand
rule: the strike line points toward the azimuth for
which the dip direction is to the right.)  The symbol
is oriented so that the long (strike) line is parallel to
the strike direction, and the symbol is labeled with
the value of the dip angle.  Special symbols are drawn
for horizontal and vertical beds (special values of
the dip angle), and for overturned beds (indicated by
a logical field in the database).  The second half of
the script, which labels the symbols with the dip val-
ue, resembles the script on the previous page, and is
not shown here.

###### Process
# Convert strike azimuth to internal coordinate system.
direction = -(azStrike - 90);
if (direction < 0) then

direction = direction + 360;
oppStrike = direction -180;
dipDir = direction - 90;
oppDip = dipDir - 180;

# Set line color, width, and end type
LineStyleSetColor(red, green, blue); # set symbol color
LineStyleSetLineWidth(lineWidth);
LineStyleSetCapJoinType(1,1); # square ends of lines

########### Draw symbol
# Special symbol for horizontal bedding (cross in circle)
if (dip1 == 0){

LineStyleDropAnchor(0);
LineStyleDrawCircle(halfLength);
LineStyleMoveTo(90, halfLength);
LineStyleLineTo(-90, strikeLength);
LineStyleMoveToAnchor(0);
LineStyleMoveTo(0, halfLength);
LineStyleLineTo(180, strikeLength);
}

else {
# For nonzero dip, draw strike line with center at point
LineStyleDropAnchor(0);
LineStyleMoveTo(direction, halfLength);
LineStyleLineTo(oppStrike, strikeLength);
LineStyleMoveToAnchor(0);

# Draw appropriate symbol for dip direction
if (dip1 == 90) { # crossbar for vertical bed

LineStyleMoveTo(dipDir, tickLength);
LineStyleLineTo(oppDip, doubTick);
}

else {
if (overturned == 1) {    # dip symbol for overturned beds

LineStyleDrawArc(0, 0, tickLength, tickLength,
direction, -180, 0);

LineStyleMoveTo(direction, tickLength);
LineStyleLineTo(oppDip, doubTick);
}

else { # dip direction tick mark
LineStyleLineTo(dipDir, tickLength);
}

}
}

Strike and Dip Script (continued)

STEPS
� click the Open

icon button on
the Display Spatial Data
toolbar and choose
Open Group from the
menu

� select the OPTGROUP

object from the TOWNS

Project File
� click the Vector

icon button for the
TOWNS layer to open the
Vector Layer Controls
window

� open the Query Editor
window for point styles
and examine the script
OPTQRY1

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Label Optimization

Label Optimization is a
procedure for automatically
finding the optimal set of
positions for point symbol
labels generated from a
database field.  The goal is
to avoid overprinting one
label with a nearby symbol’s
label.  Individual labels can
be moved or deleted to
avoid these collisions.  The
optimizer  can automatically
place a label in one of a
number of different positions
around the symbol.  The
preferred position  places the
lower left corner of the label
on the point.  Points can be
ranked using attribute
information, and these
rankings can be used by the
optimizer to give preference
to higher-ranking points
when moving or deleting
labels.

The display group used in this exercise shows a block
of counties in eastern Nebraska and the included
towns and cities.  In the script on the facing page the
1990 populations of the towns are used to assign each
to one of three rank values.  The rankings are used to
draw point symbols of different sizes, and they are
also used by the label optimizer to select labels for
deletion.

A CartoScript is executed once for each element in
an object, yet label optimization requires informa-
tion about all labels to resolve positioning conflicts.
To solve this dilemma, optimization scripts use the
LineStyleAddToOptimizer( ) function to collect
information about the dimensions of each label.  After
all points have been processed, the optimizer moves
or deletes labels as needed, then calls a function called
FuncDrawLabel( ) to draw the labels.  The
instructions for this function must be included in a
function definition in the script, as shown at the
bottom of the facing page.  This definition should
specify the font, color, and height of the label, and
include the LineStyleDrawText( ) or LineStyle-
DrawTextBox( ) function.

The first four parameters of the LineStyleAdd-
ToOptimizer( ) function are used to determine the
dimensions of a label’s bounding rectangle.  The
xstart and ystart parameters set the lower left corner
of the label, and can be read from the internal object
coordinates as shown.  The xlast and ylast parameters,
which set the upper right corner, can be calculated
from the label height and the length parameter
returned by the LineStyleTextNext-Position( )
function.  A value of 0 for the dooptimize parameter
limits changes in label position to a single pass
through the point labels.  The final dodelete parameter
is used to turn label deletion on (1) or off (0).  With
deletion on, a conflicting label of equal or lower rank
may be deleted during optimization.

# Rank towns by population
pop = TownData.POP90
if (pop < 1000) then rank = 1;
else {

if (pop >= 100000 ) then rank = 3;
else rank = 2;
}

# Draw circle with size based on rank
radius = rank * 500;
LineStyleSetColor(255,0,0);
LineStyleDrawCircle(radius,1);
# Read label text and determine dimensions
height = 3500;
LineStyleSetFont(“ARIALBD.TTF”);
LineStyleTextNextPosition(TownData.NAME,height,0,0,next_x,

next_y,length);
# Define parameters for optimization
xstart = Internal.x;        ystart  = Internal.y;
xlast = xstart + length;    ylast = ystart + height;
dooptimize = 0;             dodelete = 1;
LineStyleAddToOptimizer(xstart,ystart,xlast,ylast,rank,

dooptimize,dodelete);
# Define required function to draw labels after optimization
func FuncDrawLabel() {

LineStyleSetFont(“ARIALBD.TTF”);
LineStyleSetTextColor(0,0,0);
height = 3500;
LineStyleDrawText(TownData.NAME,height,0,0);
}

Optimization with Ranked Deletion

Display of group
with ranked
deletion.  Circles
indicate towns for
which the label
was deleted during
optimization.

There are some random aspects
to label position changes during
optimization, so redrawing the
group may cause some labels to
change position and previously
deleted labels to reappear.  The
illustration above shows one
possible set of label positions.

STEPS
� open the Query Editor

window for point styles
� choose Open / RVC

Object from the File
menu

� choose object OPTQRY2
from the TOWNS Project
File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Full Label Optimization

� close the current display
group  when you have
completed this exercise,
and click [No] on the
dialog when you are
asked whether to save
changes to the group

Full label optimization is enabled by setting the val-
ue of the dooptimize parameter in the
LineStyleAddToOptimizer( ) function to 1.  The op-
timizer then makes multiple passes through the point
labels to determine optimal label positions.  The
script on the facing page uses full optimization with-
out deletion to place labels.  It also assigns different
label sizes and colors for towns of different rank, as
illustrated below.

Note that the FuncDrawLabel( ) function cannot di-
rectly access any of the variable values assigned in
the main body of the script (including the rank vari-
able used by the optimizer).  In order to vary the
label drawing style by rank, as in this script, the
FuncDrawLabel( ) function declaration must repeat
the ranking procedure found in the main body of the
script (as well as the font assignment, source of the
label text, and other label attributes).



Using CartoScripts

page 19

# Label sizes for three sizes of towns
small = 2500; med = 3500; big = 4500;

# Rank towns by population
pop = TownData.POP90;
if (pop < 1000) then {

rank = 1; height = small;
}

else {
if (pop >= 30000 ) then {

rank = 3; height = big;
        }

else {
rank = 2; height = med;
}

}
# Draw circle with size based on rank
radius = rank * 500;
LineStyleSetColor(255,0,0);
LineStyleDrawCircle(radius,1);

# Read label text and determine dimensions
LineStyleSetFont(“ARIALBD.TTF”);
LineStyleTextNextPosition(TownData.NAME,height,0,0,next_x,

next_y,length);
# Define parameters for optimization
xstart = Internal.x; ystart  = Internal.y;
xlast = xstart + length; ylast = ystart + height;
dooptimize = 1; dodelete = 0;
LineStyleAddToOptimizer(xstart,ystart,xlast,ylast,rank,

dooptimize,dodelete);

# Define required function to draw labels after optimization
func FuncDrawLabel() {

small = 2500;   med = 3500;   big = 4500;
pop = TownData.POP90;
if (pop < 1000) {

height = small;  LineStyleSetTextColor(0,0,0);
}

else {
if (pop >= 30000 ) {

height = big;   LineStyleSetTextColor(255,0,0);
}

else {
height = med;   LineStyleSetTextColor(0,0,255);
}

}
LineStyleSetFont(“ARIALBD.TTF”);
LineStyleDrawText(TownData.NAME,height,0,0);
}

Full Optimization Script

CartoScript Shortcut
Any point symbol that you
design using the Point Style
Editor can also be saved as
a CartoScript, complete with
declared variables and
comments.  You can use
this shortcut to create the
basic script structure, then
add any custom features or
references to database
fields as needed.

Contour Line Width by Script
STEPS
� click the New 2D

Group icon
button on the Display
toolbar

� click on the Add
Vector icon button
in the Group Controls
window and choose Add
Vector Layer

� select object CONTOURS

from the CARTOSMP

Project File
� click on the Lines tab on

the Vector Layer
Controls window

� select By Script from the
Style option menu and
click [Specify...]

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object CONQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

� use the Zoom Box
tool to zoom in on
the area of closed
contours near the bottom
of the object

# ConQry
# Read contour elevation & convert to feet
elevm = Internal.MinZ;
elevft = round(elevm * 3.28084);

# Use modulo operator to
# identify contour elevations
# that are not evenly divisible
# by 200 (nonzero remainder)
rem = elevft % 200;
# Define widths for minor and
# major contours
if (rem  <> 0) then width = 2;

else width = 6;
# Set line color and width and
# draw line
LineStyleSetColor(170,85,0);
LineStyleSetLineWidth(width);
LineStyleDrawLine();

CartoScripts can also be used to draw simple or com-
plex symbols for line elements in vector or CAD
objects.  The most basic function for drawing line
symbols is the LineStyleDrawLine( ) function, which
has no parameters.  It simply draws a solid line for
each line element using the color set by the Line-
StyleSetColor( ) function and the width set by the
LineStyleSetLineWidth( ) function.  It returns the pen
position to the beginning of the line after drawing.

The script in this exercise draws lines with different
widths for major and minor elevation contours.  The
original map has a contour interval of 40 feet, and
every fifth contour (evenly divisible by 200 feet) is a
major contour shown by a wider line.  In TNTmips,
however, internal Z-values are stored in meters.  The
script reads the minimum z-value for the line from
the Internal table and converts it to feet.  The elevation
in feet is then divided by 200 using the modulo
operator, which returns the remainder of the division,
stored here in the variable rem.  The value of rem is
0 only for major contours, so this value is used to
assign the appropriate width before drawing the line.

STEPS
� reopen the Query Editor

window for line styles
� choose Open / RVC

Object from the File
menu in the Query Editor
window

� select object
STARTENDQRY from the
CARTOSMP  Project File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Navigating Lines

Start: 0

Middle: 0.5

End: 1.0

# StartEndQry
# Set radius parameters for small and
# large circles
radius1 = 8;  radius2 = 16;  dofill = 0;
# Set line color and width and draw line
LineStyleSetColor(170,85,0);
LineStyleSetLineWidth(4);
LineStyleDrawLine();
# Draw small red circle at start of line
LineStyleSetColor(225,0,0);
LineStyleDrawCircle(radius1,dofill);
# Move to end of line and draw large
# green circle
LineStyleSetPosition(1);
LineStyleSetColor(0,225,0);
LineStyleDrawCircle(radius2,dofill);

Other drawing functions can be used in conjunction
with the LineStyleDrawLine( ) function to create
more complex line symbols.  The script in this exer-
cise first draws the line element as a solid line, then
draws circles of different size and color at each end
of the line.  This script could be used for vector line
elements you are editing in the Spatial Data Editor.

A number of functions are provided to allow you to
navigate along a line element in order to draw sym-
bol components.  The LineStyleSetPosition( ) func-
tion is used in this script to move the pen position to
the end of each line after marking the start.  The
single numeric parameter of
this function specifies a line
position as a relative dis-
tance between 0 (beginning
of the line) and 1.0 (end of
the line).  You can use the
LineStyleGetPosition( )
function to find the current
pen position and assign the
value returned to a variable.
Set this function’s single pa-
rameter to 0 if you want the
returned value to be the rel-
ative position.  If you set it
to 1, the value returned is the absolute distance from
the start in object coordinates.

Relative position on line elements
using the LineStyleSetPosition( ) and
LineStyleGetPosition( ) functions.

Cartoscripts can also mark line vertices as an aid to
editing.  The script in this exercise draws a red circle
at the beginning of each line element, and a green
circle at each subsequent vertex.

The LineStyleNextVertex( ) function used in this
script moves the pen position to the next vertex along
the line.  The function also returns a value of 1 if the
end of the line has been reached, or 0 otherwise.  (The
LineStylePrevVertex( ) function moves to the previ-
ous vertex, and returns a value of 1 at the beginning
of the line.)  These functions can thus be used in a
“while” loop structure to repeat a set of drawing ac-
tions at each vertex. In this script, the “while” loop
repeats as long as the LineStyleNextVertex( ) func-
tion returns a value of 0.  The loop terminates when
the function returns a value of 1 at the end of the
line.

# VertexQry
# Set parameters for circles
# marking vertices
radius = 5;   dofill = 0;

# Draw solid black line
LineStyleSetLineWidth(3);
LineStyleSetColor(0,0,0);
LineStyleDrawLine();

# Draw red circle at beginning of line
LineStyleSetColor(225,0,0);
LineStyleDrawCircle(radius,dofill);

# While not at end of line, move to
# next vertex and draw green circle
LineStyleSetColor(0,225,0);
while (LineStyleNextVertex() <> 1) {

LineStyleDrawCircle(radius,dofill);
}

STEPS
� click on the Add

Vector icon button in the
Group Controls window
and choose Add Vector
Layer

� select object STREAMS

from the CARTOSMP

Project File
� click on the Lines tab on

the Vector Layer
Controls window

� select By Script from the
Style option menu and
click [Specify...]

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object VERTEXQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

Marking Line Vertices

� remove the STREAMS layer
when you have completed
this exercise

STEPS
� reopen the Query Editor

window for line styles for
the CONTOURS object

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object CIRCLINEQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

Drawing Regularly Spaced Symbols

# CircLineQry
# Set parameters for circles
radius = 6; dofill = 1;
# Set spacing between circles
spacing = 30;

# Set line color and width and draw line
LineStyleSetColor(170,85,0);
LineStyleSetLineWidth(3);
LineStyleDrawLine();

# Draw circle at start of line
LineStyleDrawCircle(radius,dofill);

# Draw rest of circles
while (LineStyleRoll(spacing) <> 1) {

dist = LineStyleGetDistanceTo(3);
if ( dist > spacing) {

LineStyleDrawCircle(radius,dofill);
}

}

Line symbols you create for use in map layouts may
include components spaced regularly along each line
element.  This exercise illustrates the basic structure
of such a script by drawing filled circles equally
spaced along the lines.

The LineStyleRoll( ) function moves a specified dis-
tance along a line element without drawing.  The
distance to move is set by the value of the single
function parameter.  The function also returns a val-
ue of 0 for any line position except the end, where it
returns a value of 1.  By checking this returned value
you can use the function in a “while” loop structure
to repeat a set of drawing actions at regular intervals
along each line.

This script is structured to continue drawing circles
as long as the distance from the current position to
the end of the line is greater than the desired spac-
ing.  The LineStyleGetDistanceTo( ) function is used
to check this distance.  The distance to various line
features can be determined by setting the appropri-
ate parameter value for this function, as shown in
the box to the right.

Parameter values for the
LineStyleGetDistanceTo( )
function:
1 = next vertex
2 = previous vertex
3 = end of line
4 = beginning of line

Positions and Coordinate Systems
STEPS
� reopen the Query Editor

window for line styles for
the CONTOURS object

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object TICKLINEQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

Line Start

Pointer
Position 1

Pointer
Position 2

+90 +90

-90 -90

0

0

The CartoScript drawing engine keeps track of two
positions during execution of a script.  The first is
the current position along a line element, which you
can think of as a “pointer” that is moved along the
line by the LineStyleRoll( ) function or the other line
navigation functions.  The second position that is
tracked is the pen position, which may or may not be
on the line element.

The current pointer position serves as the origin of a
local coordinate system that is oriented relative to
the line element as shown in the boxed illustration.
The drawing functions that use angle and distance
parameters to move the pen position or draw ele-

ments, such as LineStyleLineTo( ) and
LineStyleDrawArrow( ), reference this
local coordinate system.  This system
enables you to draw repeated symbol
components that are oriented consistently
relative to the local line direction, such
as the perpendicular tick lines in the
script in this exercise.  The tick lines are
drawn on the left side of each line (rela-

tive to the start and end points).  If asymmetric sym-
bols such as this need to be drawn on a particular

side of each line, you may need to use the
Spatial Data Editor to  reverse the direc-
tion of individual lines in order to create
the desired symbol.

# TickLineQry
# Denominator of
# desired map scale
scale = 5000;
# Desired spacing and
# length of tick lines
# in mm at map scale
spaceMap = 6; lengthMap = 1.5;
# Scaled spacing and length
# of tick lines
spacing = spaceMap * scale / 1000;
length = lengthMap * scale / 1000;
# Set line color and width and draw line
LineStyleSetColor(170,85,0);
LineStyleSetLineWidth(3);
LineStyleDrawLine();
# Draw tick lines
LineStyleLineTo(90,length);
while (LineStyleRoll(spacing) <> 1) {

LineStyleMoveTo(0,0);
LineStyleLineTo(90,length);
}

STEPS
� reopen the Query Editor

window for line styles for
the CONTOURS object

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object
SINEWAVEQRY from the
CARTOSMP  Project File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

When you use the LineStyleRoll( ) function, you need
to keep in mind that it moves the pointer, but not the
pen position.  In many scripts this fact is not obvious
because most of the local-reference drawing func-
tions automatically move the pen to the origin of the
local coordinate system before drawing.  In the cir-
cle-line script (page 21), for example, the LineStyle-
DrawCircle( ) function automatically moves the pen
to the current pointer position after each LineStyle-
Roll action.  One function that does not update the
pen position before drawing is the LineStyleLine-
To( ) function.  This function draws a line from the
current pen position to a point specified in
the local coordinate system.  When you want
to use this function in a LineStyleRoll loop
to draw a line beginning at the origin of the
local coordinate system, as in the tick line
script on the preceding page, use the state-
ment “LineStyleMoveTo(0,0)” to move the
pen to the current pointer position before
drawing.

The peculiarities of the LineStyleRoll( ) and
LineStyleLineTo( ) functions were engi-
neered for a purpose.  They make it possible to draw
dashed or continuous lines offset from the vector line
element.  The script in this exercise shows an ele-
gant example, a line symbol that looks like a
continuous sine curve.  The “curve” is actually made
up of small straight-line
segments drawn by the
LineStyleLineTo( ) func-
tion.  Each iteration of the
“while” loop increments
the angle for the sine func-
tion by one radian,
producing a sinusoidally
varying amplitude for the
LineStyleLineTo( ) func-
tion.

# SineWaveQry
# Set line color and width
LineStyleSetColor(0,0,0);
LineStyleSetLineWidth(3);
# Set sine wave parameters
angle = 0; space = 4;
# Draw line
while (LineStyleRoll(space) <> 1) {

angle = angle + 1; # in radians
a = sin(angle) * 5; # amplitude
if (a > 0) then LineStyleLineTo(90,a);
else LineStyleLineTo(-90,abs(a));

   }

More About Pen Position

STEPS
� reopen the Query Editor

window for line styles for
the CONTOURS object

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object BARBQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

# BarbQry
# Set line color and width
LineStyleSetColor(170,85,0)
LineStyleSetLineWidth(3)

# Set dash parameters
dashSize = 20
half = 0.5 * dashSize

# Set circle parameters
radius = 3; dofill = 1

# Draw line
while (LineStyleRoll(dashSize) <> 1) {

dist = LineStyleGetDistanceTo(3)
if (dist > dashSize) {

LineStyleRollPen(half)
LineStyleMoveTo(0,0)
LineStyleDropAnchor(0)
LineStyleLineTo(90,half)
LineStyleDrawCircle(radius,dofill)
LineStyleMoveToAnchor(0)
LineStyleRollPen(half)
}

else LineStyleRollPen(dist)
}

� remove the CONTOURS

layer when you have
completed this exercise

Drawing Dashed Lines
To draw simple or complex dashed lines, you can
use a “while” loop to alternate LineStyleRoll( ) and
LineStyleRollPen( ) actions.  The LineStyleRoll-
Pen( ) function draws a line along a line element for
a specified distance beginning at the current pointer
position.  The drawing distance is set by the value of
the single function parameter.  For a simple dashed
line with dashes and spaces of equal size, use the
same distance for both the LineStyleRoll( ) and Line-
StyleRollPen( ) functions.

The script in this exercise is a bit more elaborate,
adding a tick line with a filled circle at the end in the
middle of each dash.  In each iteration of the “while”
loop, the LineStyleRollPen( ) function draws the first
half of a dash, then the tick line and circle are drawn,
and finally the second half of the dash is drawn.

STEPS
� click on the Add

Vector icon button in the
Group Controls window
and choose Add Vector
Layer

� select object STREAMS

from the CARTOSMP

Project File
� click on the Lines tab on

the Vector Layer
Controls window

� select By Script from the
Style option menu and
click [Specify...]

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object
DOUBDASHQRY from the
CARTOSMP  Project File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

Double Dashed Lines
The script in this exercise illustrates another dashed
line variation.  It draws each line with double dashes
connected by crossing lines.  Each pair of dashes is
centered on the line element, which means that the
dashes themselves are offset from the line by a dis-
tance specified by the offset variable.  Because no
part of the symbol traces the line element itself, the
LineStyleLineTo( ) function is used to draw both the
dashes and the crossing lines.

# DoubDashQry
dashSize = 15;
halfDash = dashSize * 0.5;
double = 2 * dashSize;
offset = dashSize * 0.2;
doubOffset = offset * 2;

# Set line color and width
LineStyleSetColor(255,0,0);
LineStyleSetLineWidth(2);

# Draw double dash line and crossing lines
LineStyleMoveTo(90, offset);
LineStyleLineTo(0, dashSize);
LineStyleMoveTo(-90, doubOffset);
LineStyleLineTo(180, dashSize);
LineStyleMoveTo(0, halfDash);
LineStyleLineTo(90, doubOffset);

while (LineStyleRoll(double) <> 1) {
dist = LineStyleGetDistanceTo(3);
if (dist > dashSize) {

LineStyleMoveTo(90, offset);
LineStyleLineTo(0, dashSize);
LineStyleMoveTo(-90, doubOffset);
LineStyleLineTo(180, dashSize);
LineStyleMoveTo(0, halfDash);
LineStyleLineTo(90, doubOffset);
}

}

Line symbols can include components repeated at
different intervals along the line, as illustrated by
the dashes and diamonds drawn by the script in this
exercise.  The variable cumL in the script keeps track
of the cumulative length of dashes and spaces drawn
since the last diamond symbol.  When the value of
cumL reaches the spacing distance set for the
diamonds, a diamond symbol is drawn instead of a
dash, and the value of cumL is reset to 0.  The
diamond symbol is created here using the Line-
StyleSideshot( ) function.  This function allows you
to specify a number of points by angle and distance
in the local coordinate system, and connect them to
form a polyline by setting the value of the dodraw
parameter to 1.  This script also records the points
as a polygon so the diamond shape can be filled.

Handling Multiple Repeat Intervals

# DashDiQry
# Set line color and width
LineStyleSetColor(225,0,0);
LineStyleSetLineWidth(2);
# Set dash parameters
dashSize = 12; half = 0.5 * dashSize;
# Set diamond parameters
spacing = 5 * dashSize;
dodraw = 1; dofill = 1;
width = 0.3 * dashSize;
cumL = 0; # Cumulative length variable
# Draw line
while (LineStyleRoll(dashSize) <> 1) {

cumL = cumL + dashSize; # increment cumulative length
if (cumL >= spacing) { # draw diamond symbol

LineStyleRoll(half);
LineStyleMoveTo(0,0);
LineStyleRecordPolygon(1);
LineStyleSideshot(dodraw,0,half,90,width,180,

half,-90,width);
LineStyleDrawPolygon(dofill);
LineStyleRoll(half);
cumL = 0; # reset cumulative length to 0
}

else {
LineStyleRollPen(dashSize); # draw dash
cumL = cumL + dashSize; # increment cumulative length
}

}

STEPS
� reopen the Query Editor

window for line styles for
the STREAMS object

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object DASHDIQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

# DashThrustQry
dashSize = 16; # length of dashes
halfDash = dashSize * 0.5;
# triangle spacing and dimensions
spacing = dashSize * 6 ;
quartSpace = spacing * 0.25;
triWidth = dashSize;
halfTri = triWidth * 0.5;
height = dashSize * 0.6;
# Set line color and width
LineStyleSetColor(255,0,0);
LineStyleSetLineWidth(2);
# Initialize variable to control placement of triangles
cumL = spacing;
# Draw dashed fault line and triangles
while (LineStyleRoll(halfDash) <> 1) {

dist = LineStyleGetDistanceTo(3);   # distance to end of line
cumL = cumL + dashSize;
if (dist > quartSpace and cumL >= spacing) {

LineStyleDropAnchor(1);
LineStyleRoll(halfTri);
LineStyleMoveTo(0,0);
LineStyleMoveTo(90, height);
LineStyleDropAnchor(2);
LineStyleRoll(-halfTri);
LineStyleRecordPolygon();
LineStyleRollPen(triWidth);
LineStyleLineToAnchor(2);
LineStyleLineToAnchor(1);
LineStyleDrawPolygon(1);
cumL = 0;
}

else {
LineStyleRollPen(dashSize);
cumL = cumL + dashSize;
}

}

Dashed Thrust Fault Symbol
STEPS
� reopen the Query Editor

window for line styles for
the STREAMS object

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object
DASHTHRUSTQRY from the
CARTOSMP  Project File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

The script in this query provides another example of
symbols using different repeat intervals.  It draws
the dashed version of the geologic map symbol for a
thrust fault.  The triangles are drawn as polygons
using anchor points placed at the corners.  The bases
of the triangle symbols are drawn using the Line-
StyleRollPen( ) function so that they conform to
curves in the line elements.  Note that a negative
distance can be used with the LineStyleRoll( ) func-
tion to move backward along a line element.

� remove the STREAMS

layer when you
have completed this
exercise

Labeling Contour Lines
STEPS
� click on the Add

Vector icon button in the
Group Controls window
and choose Add Vector
Layer

� select object CONTOURS

from the CARTOSMP

Project File
� click on the Lines tab on

the Vector Layer
Controls window

� select By Script from the
Style option menu and
click [Specify...]

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object CONLABLQRY

from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

You can design a CartoScript to vary the placement
and orientation of line symbol components to
accomodate the local direction and shape of the line
elements.  In this script that draws and labels con-
tours, the elevation labels for the major contours are
moved if the local portion of the line is too highly
curved, and are inverted if necessary to be readable.
(Only the main processing portion of the script is
shown on the next page).

These conditions are checked using the Line-
StyleGetDirection( ) and LineStyleGetMaxDis-
tance( ) functions.  The first parameter of both
functions is a value that specifies the length of the
portion of the line that you want to examine.  The
additional parameters of both functions are variables
that are assigned values by the function.  The Line-
StyleGetDirection( ) function finds the minimum and
maximum direction angles for the specified line seg-
ment in the object coordinate reference frame (pos-
itive x-axis = 0 degrees).  The LineStyleGet-
MaxDistance( ) function finds the maximum perpen-
dicular distance between the specified portion of the
line element and a straight line joining its endpoints
(see illustration at left), as well as the direction an-
gle of this straight line (used in this script to orient
the contour label).  Both functions return a value of
1 if at the end of the line element, or 0 otherwise.
The script checks the maximum distance value as

well as the change in line di-
rection over the label length
to determine if the local por-
tion of the line is too highly
curved to place the label
there.  The direction angle of
the label line is used to iden-
tify labels that need to be in-
verted.

maxDist from
LineStyleGet-
MaxDistance( )

Current
line
position

Straight line joining endpoints

� remove the CONTOURS

layer when you have
completed this exercise

Labeling Contour Lines (continued)
if (rem <> 0) { # draw minor contours

LineStyleSetLineWidth(width);
LineStyleDrawLine();
}

else { # draw and label major contours
LineStyleSetLineWidth(widthBold);
str$ = sprintf(“%d”,elev); # read elevation to string variable
# find length of contour label
LineStyleTextNextPosition(str$,labelSize,0,1,nextx,

nexty,length);
begShift = 5 * length;    # offset from beginning of line
stopLength = begShift;    # min label distance from end of line
spLength = 1.5 * length;   # label length plus spaces

LineStyleRollPen(begShift); # draw beginning of contour line

while ((LineStyleGetMaxDistance(spLength,drawAngle,
maxDist)) <> 1) {

LineStyleGetDirection(0.1 * labelSize,minAngle,maxAngle);
# find change in line direction over length of label
devAngle = drawAngle - minAngle;
# find distance to end of line and compare to stopLength
remLength = LineStyleGetDistanceTo(3);
if (remLength < stopLength) break;
# check deviation distance and angle of line segment
if ((maxDist < 1.5 * labelSize) and (abs(devAngle) < 15)) {

LineStyleRoll(0.25*length); # space before drawing label

# Check if label needs to be inverted to be readable
isInverse = 0;
if (abs(drawAngle) > 90) isInverse = 1;
if (isInverse) {

LineStyleMoveTo(devAngle, length);
LineStyleMoveTo(devAngle + 90, halfSize);
LineStyleDrawText(str$,labelSize,drawAngle+180,1);
}

else {
LineStyleMoveTo(-90,halfSize);
LineStyleDrawText(str$,labelSize,drawAngle,1);
}

LineStyleRoll(1.2 * length);
LineStyleMoveTo(0,0);
LineStyleSetLineWidth(widthBold);
LineStyleRollPen(minDistBetweenLabels);
}

else LineStyleRollPen(length);
}

# Draw remainder of line
LineStyleRollPen(remLength);
}

STEPS
� click on the Add

Vector icon button in the
Group Controls window
and choose Add Vector
Layer

� select object SYNCLINE

from the CARTOSMP

Project File
� click on the Lines tab on

the Vector Layer
Controls window

� select By Script from the
Style option menu and
click [Specify...]

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object
SYNCLINEQRY from the
CARTOSMP  Project File

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

# SynclineQry
# Read a logical database field (Yes/No) to check if syncline is
# overturned.  Numeric variable is set to 1 if yes, 0 if no
overturned = Syncline.Overturned;
# dimensions of arrow symbols
arrowSize = 30; halfSize = arrowSize * 0.5;
headSize = 0.5 * arrowSize; sweepAngle = 45;
stemSize = arrowSize - headSize * cosd(sweepAngle);
width = 2; # width of lines
# Parameters for half circle in overturned syncline symbol
angle = 0; shift = halfSize; # angle and distance to arc center
radius_x = halfSize; # radii of arc
radius_y = halfSize;
startAngle = -180; swpAngle = -180;
rotAngle = 0; isAngleAbs = 0;
# Set line color, width, and draw fold line
LineStyleSetColor(228,0,0);
LineStyleSetLineWidth(width);
LineStyleDrawLine();

Syncline Line Symbol
A CartoScript can be structured to draw different line
symbols depending on the attributes attached to the
line elements.  The script in this exercise draws geo-
logical line symbols for the axial trace of a synclinal
(downward) fold in layered rocks.  If one limb of the
fold has rotated beyond a vertical orientation, the fold
is overturned, and a special symbol is used.  The
overturned condition for vector lines in this exercise
is indicated by a logical database field.

This script draws each line element as a solid line,
then places the fold symbol in the middle of the line.
The semicircle that forms part of the overturned
syncline symbol is drawn using the LineStyleDraw-
Arc( ) function, which draws an arc about a specified
center point.  The first two function parameters
specify an angle and distance to the intended center
point of the arc, so you have the option of moving
the pen to a new location before drawing the arc.
You also specify a starting angle and sweep angle
for the arc, and have the option of rotating the entire
arc after drawing by assigning a nonzero value for
the rotangle parameter.

# Draw arrow symbols in middle of each line
LineStyleSetPosition(0.5); # move to middle of line
 if ( overturned == 1 ) { # draw overturned syncline symbol

# Draw arrows with 0 line width
LineStyleRoll(-halfSize);
LineStyleDropAnchor(1);
LineStyleSetLineWidth(0);
LineStyleMoveTo(-90, arrowSize);
LineStyleDropAnchor(2);
LineStyleDrawArrow(90, arrowSize, headSize, sweepAngle, 1);
LineStyleMoveTo(0, arrowSize);
LineStyleMoveTo(-90, arrowSize);
LineStyleDropAnchor(3);
LineStyleDrawArrow(90, arrowSize, headSize, sweepAngle, 1);

# Draw arrow stems and arc with linewidth = width
LineStyleSetLineWidth(width);
LineStyleMoveToAnchor(2);
LineStyleLineTo(90, stemSize);
LineStyleMoveToAnchor(1);
LineStyleDrawArc(angle, shift, radius_x, radius_y,

 startAngle, swpAngle, isAngleAbs);
LineStyleMoveToAnchor(3);
LineStyleLineTo(90, stemSize);
}

else { # draw upright syncline symbol
# Draw arrows with 0 line width
LineStyleDropAnchor(1);
LineStyleSetLineWidth(0);
LineStyleMoveTo(90, arrowSize);
LineStyleDropAnchor(2);
LineStyleDrawArrow(-90, arrowSize, headSize, sweepAngle, 1);
LineStyleMoveToAnchor(1);
LineStyleMoveTo(-90, arrowSize);
LineStyleDropAnchor(3);
LineStyleDrawArrow(90, arrowSize, headSize, sweepAngle, 1);

# Redraw arrow stems
LineStyleMoveToAnchor(2);
LineStyleSetLineWidth(width);
LineStyleLineTo(-90, stemSize);
LineStyleMoveToAnchor(3);
LineStyleLineTo(90, stemSize);
}

Syncline Line Symbol (continued)

This script draws the dip arrows
for the overturned syncline
symbol on the right side of a
line element.  Line directions
have been set in the vector
object  to accomodate this
script design.

Add Elements to LegendView
STEPS
� make sure that lines are

selectable in the View
window

Vector style samples appear automatically in Leg-
endView when you style points, lines, or polygons
using the All Same, By Attibute, or By Theme style

options.  But when you style vector elements us-
ing a script, the “styles” created by your script do
not automatically appear in LegendView; instead
the current All Same style for the element ap-

pears in LegendView by default.

You can create LegendView samples for vector ele-
ments styled by script by selecting from the View
window a representative element for each drawing
“style” and adding it to LegendView.  You must use
the selection tools in the GeoToolbox to perform the
selection (you cannot use the simple Select tool on
the View window).

In this exercise you select two lines from the syn-
cline vector layer, one styled with the normal upright
syncline symbol and the other with the overturned

syncline symbol.  When each LegendView sam-
ple is rendered in the window, the script is
executed using the database attributes of the ele-
ment you selected for it.

� select the
GeoToolbox from
the View window

� turn on the Select
tool in the
GeoToolbox
window

� in the View window, left-
click on one of the two
lines drawn with the
upright syncline symbol

� press and hold the right
mouse button and select
Add Active Line to
Legend from the popup
menu

� when the Legend
Element Label window
appears, enter “Syncline”
in the text field and press
[OK]

� repeat the last three
steps for the line drawn
with the overturned
syncline symbol, entering
“Overturned Syncline” for
the label

A special database table is created to identify the
elements you have selected for the legend.  The
table also stores the label text for each sample.

STEPS
� open the Vector Layer

Controls window and
click on the Lines tab

� select Straight from the
Legend Style menu on
the lower part of the
panel

� reopen the Query Editor
window for line styles

� choose Open / RVC
Object from the File
menu in the Query Editor
window

� select object SynQryLgn
from the CARTOSMP

Project File
� click [OK] in the Query

Editor window and again
in the Vector Layer
Controls window

 Script Variables for Legend Samples

# These variables control the length of the arrows
# ArrowLengthMap is the desired arrow length in mm,
# assuming vector coordinates are in meters:
if (DrawingLegendView == 1) {

ArrowSize = 0.5 * SampleRect.GetHeight();
}
else {

ArrowLengthMap = 4;
ArrowSize = ArrowLengthMap * Scale / 1000;

}

The default shape for a line sample in
LegendView is zigzag.  Complex line
symbols drawn by CartoScript look
better in LegendView when drawn with
the Straight legend style option.

Notice that the syncline symbols drawn in Legend-
View in the previous exercise overlap each other
because they appear too large for the legend spac-
ing.  LegendView assigns a rectangular area to each
sample, and the general scaling parameters included
in your script for map rendering may not provide a
good fit to that area.

Several script variables let you set up special pro-
cessing to better scale symbols for legends.  When
the Display process renders a sample in LegendView
using a script, the variable DrawingLegendView is
automatically assigned a value of 1.  (Its value is 0
when the script is rendering elements in the view.)
You can therefore set up a conditional loop in the
script that checks this value and executes only when
the script is drawing a sample for LegendView.

When a legend sample is being rendered,
class SampleRect is created automatically
to allow you to access the coordinates and
size of the sample rectangle.  SampleRect
is an instance of the general class Rect,
which includes methods GetHeight() and
GetWidth() that you can use to find either
dimension of the rectangle.  The Carto-
Script in this exercise uses the height of
the sample rectangle to scale the syncline
symbols in LegendView.  The relevant
portion of the script is excerpted below.

� close the
current display
group  when
you have
completed this
exercise

CartoScript Samples in Printed Legends
STEPS
� click the Open

icon button on
the Spatial Data Display
toolbar and choose
Open Layout from the
menu

� select  PRINTLAYOUT from
the GGMAP Project File

� in the Layout
Controls window,
click on the
Legend icon button for
the Symbols legend

� examine the Legend
Layer Controls window,
then click [OK]

Once you have selected legend samples for vector
elements that are being rendered by a script, you can
also create a legend with these symbols in a map
layout for printing.  Simply create a multi-object leg-
end, use the Legend Layer Control window to add
the appropriate element type (point, line, or poly-
gon) from the vector object, and the script-styled
legend samples and their labels appear automatical-
ly in the legend.  The Legend Layer Controls window
allows you to adjust the positions of legend entries
and to change the label text and style if desired.  The
tutorial booklet Making Map Layouts includes sev-
eral exercises to guide you through the creation of a
multi-object legend.

The print layout you open in this exercise in-
cludes a multi-object legend of geologic line
and point symbols rendered by CartoScript.
All of the script objects used are also includ-
ed in the Project File.

STEPS
� in the Layout Controls

window, open the Vector
Layer Controls window
for the Lineation vector
layer in Group 1

� click on the Points panel
in the Vector Layer
Controls window and
open the Query Editor
window for point styles

� note how legend
variables are used in the
script

� click [OK] in the Query
Editor window and again
in the Vector Layer
Controls window

# Draw arrow with zero line width, tip of stem at data point
# but centered on point if legend sample
if (DrawingSample == 1) {

LineStyleMoveTo(OppTrend, ArrowLength / 2);
LineStyleMoveTo(-90,5);

}

LineStyleDropAnchor(0); # anchor at data point
LineStyleSetLineWidth(0);
LineStyleDrawArrow(Direction, ArrowLength, HeadSize, Angle, 1);
LineStyleDropAnchor(1); # anchor at tip of arrow

# Draw arrow stem with desired line width
LineStyleMoveToAnchor(0);
LineStyleSetLineWidth(LineWidth);
LineStyleLineTo(Direction, StemLength);
LineStyleMoveToAnchor(1);

Scripting  Legend Sample Position
Point and line symbols should be rendered in a printed
legend at the same scale as they are on the map.  If
you have scaled your symbols to a map scale, they
will appear at the correct scale automatically in a
multi-object legend.  But there are instances in which
you might want to have the script adjust the position
of samples both in LegendView and in a multi-ob-
ject legend.  The script variable DrawingSample is
automatically set to 1 whenever the script renders
either type of legend sample, allowing you to set up
instructions in the script to reposition symbols in the
legend sample.

The script in this exercise is a version of one shown
in the earlier exercise Orienting Symbols by At-
tribute.  It draws labeled arrows that in this case show
the trend and plunge of mineral lineations at various
map locations.  In the map the base of the arrow is at
the outcrop location.  The drawing instruc-
tions are modified when a legend sample
is being rendered to center the arrow on the
point and move it down a few pixels to bet-
ter align vertically with the legend label.
The relevant portion of the script is excerpt-
ed below.

CartoScript Function List
Style Control Functions
LineStyleSetCapJoinType

LineStyleSetColor
LineSyleSetCoordType

LineStyleSetFont
LineStyleSetLineWidth
LineStyleSetScale
LineStyleSetTextColor

Label Optimization
LineStyleAddToOptimizer

Functions that reference position within input line elements
Navigation

LineStyleNextVertex
LineStylePrevVertex
LineStyleRoll

LineStyleSetPosition

Drawing
LineStyleRollPen

Information
LineStyleGetDirection

LineStyleGetDistanceTo

LineStyleGetLineCurvature
LineStyleGetMaxDistance

LineStyleGetPosition

LineStyleIsClosed

Note: the line segment operated on by certain functions in the last group above
begins at the current pointer position along the line element and extends to a
specified distance from it.

set square or rounded ends for lines and polyline
segments
set RGB values (0-255) for line color
set coordinate type for input values (object or
millimeters)
set name of font to use for text
set width of lines to be drawn
set scale factor applied to input distances
set RGB values (0-255) for text color

move pen to next vertex along line element
move pen to previous vertex along line
move pointer a specified distance along line
element
move pen to specified relative position along
line element (0.0 = start, 1.0 = end)

move pen to pointer position and draw line for
specified distance along line element

find minimum and maximum direction angles
of line element segment
find distance from pointer position to next
vertex (1), previous vertex (2), end (3), or start
(4) of line element
find curvature of line element segment
find maximum perpendicular distance from
line segment to straight line connecting
segment start and end
find current relative position of pointer along
line element (0.0 = start, 1.0 = end)
test if current line element forms a closed
polygon (1) or not (0)

access built-in label placement optimizer

CartoScript Function List (continued)
Functions that reference the current local coordinate system

Navigation
LineStyleDropAnchor

LineStyleMoveTo

LineStyleMoveToAnchor

Drawing
LineStyleDrawArc
LineStyleDrawArrow

LineStyleDrawCircle
LineStyleDrawCone
LineStyleDrawCube
LineStyleDrawCylinder
LineStyleDrawEllipse
LineStyleDrawPolygon

LineStyleDrawPolyline

LineStyleDrawRectangle
LineStyleDrawText
LineStyleDrawTextBox
LineStyleDrawThreePointArc
LineStyleLineTo

LineStyleLineToAnchor

LineStyleRecordPolygon

LineStyleSideshot

LineStyleTextNextPosition

Functions that draw or transform entire input line elements
LineStyleDrawLine
LineStyleRestoreLine
LineStyleSpline
LineStyleThinLine

record current pen position as anchor
number for later use
move pen to location specified by direction
and distance from current pen location
move pen to specified anchor location

draw arc
draw arrow to location specified by direction
and distance from current pen location
draw filled or unfilled circle
draw cone
draw cube
draw cylinder
draw filled or unfilled ellipse
draw polygon connecting previously
recorded points
draw polyline connecting previously
recorded points
draw filled or unfilled rectangle
draw text label with specified text string
draw boxed text label
draw arc connecting three points
draw line to location specified by direction
and distance from current pen location
draw line from current pen position to
specified anchor location
start or stop recording polygon vertex
locations
specify sequence of positions by direction
and distance from current pen position and
optionally connect to form polyline
compute length and end position of text
string

draw entire vector or CAD line element
restore original line coordinates
replace line element with splined line
replace line element with thinned line

Advanced Software for Geospatial Analysis C
A
R
T
O
S
C
R
I
P
T
S

M icro Im a g es, In c .
11th Floor - Sharp Tower
206 South 13th Street
Lincoln, Nebraska  68508-2010  USA

Voice: (402) 477-9554 email: info@microimages.com
FAX: (402) 477-9559 internet: www.microimages.com

MicroImages, Inc. publishes a complete line of professional software for advanced geospatial
data visualization, analysis, and publishing.  Contact us or visit our web site for detailed prod-
uct information.

TNTmips TNTmips is a professional system for fully integrated GIS, image analysis, CAD,
TIN, desktop cartography, and geospatial database management.

TNTedit TNTedit provides interactive tools to create, georeference, and edit vector,
image, CAD, TIN, and relational database project materials in a wide variety of formats.

TNTview TNTview has the same powerful display features as TNTmips and is perfect for
those who do not need the technical processing and preparation features of TNTmips.

TNTatlas TNTatlas lets you publish and distribute your spatial project materials on CD-
ROM at low cost.  TNTatlas CDs can be used on any popular computing platform.

TNTserver TNTserver lets you publish TNTatlases on the Internet or on your intranet.
Navigate through geodata atlases with your web browser and the TNTclient Java applet.

TNTlite TNTlite is a free version of TNTmips for students and professionals with small
projects.  You can download TNTlite from MicroImages’ web site, or you can order
TNTlite on CD-ROM.

